Solving The Exponential Integral
Excel VBA Toolbox

Gary Schurman, MBE, CFA

December, 2024

The exponential integral (via the incomplete gamma function) allows us to transition company economics from one state to another over time. The mechanics are that a short-term unsustainable rate transitions to a long-term sustainable rate via mean reversion.

Example: ABC Company is growing its revenues at a rate well in excess of nominal GDP. This revenue growth rate is not sustainable and as the company matures that revenue growth rate should decrease over time to approximate nominal GDP minus one or two hundred basis points (rule of thumb). Given a constant after-tax revenue margin, net income should be ever increasing (first derivative is positive) but at a decreasing rate (second derivative is negative).

Our Hypothetical Problem

We are currently standing at time zero and are tasked with estimating ABC Company's future revenue. Our go-forward model assumptions are...

Description	Value
Annualized revenue in year zero (in dollars)	1,000,000
Annual revenue growth rate - Short-term rate (%)	20.00
Annual revenue growth rate - Long-term rate (%)	3.50
After-tax revenue margin (%)	25.00
Risk-adjusted discount rate (%)	10.00
Transition half-life (in years)	5.00

Question 1: What is the continuous-time revenue growth rate at the end of year 5?

Question 2: What is annualized revenue at the end of year 5?

Question 3: What is after-tax revenue margin recognized in year 5?

Question 4: What is the present value of after-tax revenue margin recognized in year 5?

The Mean-Reverting Revenue Growth Rate [1]

We will define the variable λ to be the rate of mean reversion, which is the rate at which the short-term rate transitions to the long-term rate over time. The equation for the periodic rate at time t from the perspective of time zero is...

$$\text{Rate at time } t = \text{Long-term rate} + \left(\text{Short-term rate} - \text{Long-term rate}\right)\text{Exp}\left\{-\lambda\, t\right\} \tag{1}$$

Note that the limit of Equation (1) above as time goes to infinity is...

$$\lim_{t\to\infty} \text{Rate at time } t = \text{Long-term rate} \quad \text{...because...} \quad \lim_{t\to\infty} \text{Exp}\left\{-\lambda\, t\right\} = 0 \tag{2}$$

In the equations above we defined the variable λ to be the rate of mean reversion. To calibrate λ we will choose some future point in time (time $= T$) where the periodic rate is halfway between the short-term rate and the long-term rate (i.e. the transition half-life). The equation to calibrate λ is therefore...

$$\text{if... } \mathrm{Exp}\left\{-\lambda \times T\right\} = 0.50 \text{ ...then... } \lambda = -\frac{\ln(0.50)}{T} \tag{3}$$

Using the VBA Function LambdaCalc() below, the rate of mean reversion for our problem is...

$$\lambda = \mathrm{LambdaCalc}(5.00) = 0.1386 \tag{4}$$

We want our revenue growth rates to be continuous-time rates. We will define the variable ω_S to be the short-term, continuous-time revenue growth rate, the variable ω_L to be the long-term, continuous-time revenue growth rate, and the variable ω_t to be the revenue growth rate at time t. Using the VBA Function DTOCCalc() below, the equations for our continuous-time rates are...

$$\omega_S = \mathrm{DTOCCalc}(0.2000) = 0.1823 \text{ ...and... } \omega_L = \mathrm{DTOCCalc}(0.0350) = 0.0344 \tag{5}$$

We will define the variable Γ_t to be the cumulative revenue growth rate over the time interval $[0,t]$. Using Equations (4) and (5) above, the equation for the cumulative revenue growth rate at time t is...

$$\Gamma_t = \int\limits_0^t \omega_s \,\delta s = \frac{\Delta(\omega)}{\lambda} + \omega_L\, t - \frac{\Delta(\omega)}{\lambda}\,\mathrm{Exp}\left\{-\lambda\, t\right\} \text{ ...where... } \Delta(\omega) = \omega_L - \omega_S \tag{6}$$

Question 1: What is the continuous-time revenue growth rate at the end of year 5?

Using the VBA Function MRRate() below and our model parameters above, the answer to the question is...

$$\omega_5 = \mathrm{MRRate}(0.1386,\ 0.1823,\ 0.0344,\ 5) = 0.1084 \tag{7}$$

Annualized Revenue [2]

We will define the variable R_t to be annualized revenue at time t. Using Equation (6) above, the equation for annualized revenue is...

$$R_t = R_0\,\mathrm{Exp}\left\{\Gamma_t\right\} = R_0\,\mathrm{Exp}\left\{\frac{\Delta(\omega)}{\lambda} + \omega_L\, t - \frac{\Delta(\omega)}{\lambda}\,\mathrm{Exp}\left\{-\lambda\, t\right\}\right\} \tag{8}$$

To simplify the calculations that follow, we will make the following equation definitions...

$$E_1 = \mathrm{Exp}\left\{\frac{\Delta(\omega)}{\lambda} + \omega_L\, t - \frac{\Delta(\omega)}{\lambda}\,\mathrm{Exp}\left\{-\lambda\, t\right\}\right\} \tag{9}$$

$$E_2 = \mathrm{Exp}\left\{\frac{\Delta(\omega)}{\lambda} + (\omega_L - \lambda)\, t - \frac{\Delta(\omega)}{\lambda}\,\mathrm{Exp}\left\{-\lambda\, t\right\}\right\} \tag{10}$$

$$E_3 = \mathrm{Exp}\left\{\frac{\Delta(\omega)}{\lambda} + (\omega_L - 2\,\lambda)\, t - \frac{\Delta(\omega)}{\lambda}\,\mathrm{Exp}\left\{-\lambda\, t\right\}\right\} \tag{11}$$

Using Equation (9) above, we can rewrite Equation (8) above as...

$$R_t = R_0\, E_1 \tag{12}$$

Question 2: What is annualized revenue at the end of year 5?

Using the VBA Function BaseEquation() below and our model parameters above, the answer to the question is...

$$R_5 = 1{,}000{,}000 \times \mathrm{BaseEquation}(1,\ 0.1386,\ 0.1823,\ 0.0344,\ 5) = 2{,}024{,}892 \tag{13}$$

Net Income/(Loss)

We will define the variable θ to be the after-tax revenue margin. Using Equation (12) above and our model parameters above, the value of the model parameter θ is...

$$\theta = 0.2500 \tag{14}$$

We will define the variable N_t to be annualized after-tax revenue margin at time t. Using Equation (12) above, the equation for annualized net income at time t is...

$$N_t = \theta\, R_t = \theta\, R_0\, E_1 \tag{15}$$

Using Equation (12) above and our model parameters above, the equation for cumulative nominal net income is...

$$N_{m,n} = \int\limits_n^m \theta\, R_0\, E_1\, \delta t = \theta\, R_0 \int\limits_n^m E_1\, \delta t \tag{16}$$

Note that there is not closed-form soluation to Equation (16) above and therefore must be solved via numerical integration.

Question 3: What is after-tax revenue margin recognized in year 5?

Using the VBA Function NetIncome() below and our model parameters above, the answer to the question is...

$$N_{4,5} = \text{NetIncome}(1000000,\ 0.1386,\ 0.1823,\ 0.0344,\ 0.2500,\ 4.00,\ 5.00) = 480,296 \tag{17}$$

PV of Net Income/(Loss)

We want our risk-adjusted discount rates to be a continuous-time rate. We will define the variable κ to be the continuous-time, risk-adjusted discount rate. Using the VBA Function DTOCCalc() below, the equation for our continuous-time discount rate is...

$$\kappa = \text{DTOCCalc}(0.1000) = 0.0953 \tag{18}$$

The discounted versions of Equations (9), (10), and (11) above are...

$$E_{1a} = E_1\, \text{Exp}\left\{ -\kappa\, t \right\} = \text{Exp}\left\{ \frac{\Delta(\omega)}{\lambda} + (\omega_L - \kappa)\, t - \frac{\Delta(\omega)}{\lambda}\, \text{Exp}\left\{ -\lambda\, t \right\} \right\} \tag{19}$$

$$E_{2a} = E_2\, \text{Exp}\left\{ -\kappa\, t \right\} = \text{Exp}\left\{ \frac{\Delta(\omega)}{\lambda} + (\omega_L - \lambda - \kappa)\, t - \frac{\Delta(\omega)}{\lambda}\, \text{Exp}\left\{ -\lambda\, t \right\} \right\} \tag{20}$$

$$E_{3a} = E_3\, \text{Exp}\left\{ -\kappa\, t \right\} = \text{Exp}\left\{ \frac{\Delta(\omega)}{\lambda} + (\omega_L - 2\,\lambda - \kappa)\, t - \frac{\Delta(\omega)}{\lambda}\, \text{Exp}\left\{ -\lambda\, t \right\} \right\} \tag{21}$$

We will define the variable $\bar{N}_{m,n}$ to be the present value of cumulative nominal net income recognized over the time interval $[m, n]$. Using Equation (19) above, the discounted version of Equation (16) above

$$\bar{N}_{m,n} = \int\limits_n^m \theta\, R_0\, E_1\, \text{Exp}\left\{ -\kappa\, t \right\} \delta t = \theta\, R_0 \int\limits_n^m E_{1a}\, \delta t \tag{22}$$

Question 4: What is the present value (at time zero) of after-tax revenue margin recognized in year 5?

Using the VBA Function EXPINT() below and our model parameters above, the answer to the question is...

$$\bar{N}_{m,n} = 0.25 \times 1,000,000 \times \text{EXPINT(1,0.0953,0.1386,0.1823,0.0344,4,5)} = 311,732 \tag{23}$$

References

[1] Gary Schurman, *The Stochastic, Mean-Reverting Short Rate,* May, 2020

[2] Gary Schurman, *Incomplete Gamma Function - Base Equation For A Mean-Reverting Process,* November, 2017

The VBA functions for the Exponential Integral toolbox are included below...

```
'Name: Toolbox
'Purpose: Exponential integral solution toolbox.
'Author: Gary Schurman, MBE, CFA

Option Explicit

'Name: NetIncome
'Purpose: Returns nominal net income over the time interval [m,n].
Public Function NetIncome(baseRevenue As Double, lambda As Double, omega_st As Double, _
omega_lt As Double, theta As Double, m As Double, n As Double) As Double
    'Declare calculation variables.
    Dim mValue As Double: Dim mValue01 As Double
    Dim delta As Double
    Dim mCount As Double
    Dim mTime As Double
    'Set step size variable value.
    delta = 0.05
    'Calculate return variable value.
    For mCount = 1 To 1 / delta
        'Calculate current time in years.
        mTime = m + delta * mCount
        'Get net income for this subinterval.
        mValue01 = baseRevenue * theta * BaseEquation(1, lambda, omega_st, omega_lt, mTime) * delta
        'Accumulate results.
        mValue = mValue + mValue01
    Next mCount
    'Return value to caller.
    NetIncome = mValue
End Function

'************************************
'Base Equations
'************************************

'Name: BaseEquation
'Purpose: Returns equations E1 (version:1), E2 (version:2) and E3 (version:3).
Public Function BaseEquation(version As Integer, lambda As Double, omega_st As Double, omega_lt As Double,
    'Declare calculation variables.
    Dim mValue As Double
    Dim delta_r As Double
    'Define calculation variable values.
    delta_r = omega_st - omega_lt
    mValue = Exp(delta_r / lambda + (omega_lt - lambda * (version - 1)) * t - delta_r / lambda * Exp(-lamb
    'Return value to caller.
    BaseEquation = mValue
End Function

'Name: BaseEquationPV
'Purpose: Returns equations E1a (version:1), E2a (version:2) and E3a (version:3).
Public Function BaseEquationPV(version As Integer, kappa As Double, lambda As Double, omega_st As Double,
omega_lt As Double, t As Double) As Double
    'Declare calculation variables.
    Dim mValue As Double
    Dim delta_r As Double
```

```
    'Define calculation variable values.
    delta_r = omega_st - omega_lt
    mValue = Exp(delta_r / lambda + (omega_lt - lambda * (version - 1) - kappa) * t - delta_r / lambda * E
    'Return value to caller.
    BaseEquationPV = mValue
End Function

'Name: EXPINT
'Purpose: Returns exponential integral value.
'Note: Version 1: c = omega - kappa Version 2: c = omega - lambda - kappa Version 3: c = omega - 2 * lambd
'Note: If n = 0 then the integral upper bound is assumed to be infinity.
Public Function EXPINT(version As Integer, kappa As Double, lambda As Double, omega_st As Double, omega_lt
m As Double, n As Double) As Double
    'Declare calculation variables.
    Dim mValue As Double: Dim mValue01 As Double
    Dim xm As Double: Dim xn As Double
    Dim alpha As Double
    'Declare and define c variable.
    Dim CVariable As Double
    Select Case version
        Case 1: CVariable = omega_lt - kappa
        Case 2: CVariable = omega_lt - lambda - kappa
        Case 3: CVariable = omega_lt - 2 * lambda - kappa
    End Select
    'Declare and define delta variable.
    Dim delta As Double
    delta = omega_st - omega_lt
   'Define return variable value.
    If delta > 0 Then
        'Declare incomplete gamma integral parameters.
        Dim a As Double: Dim b As Double: Dim c As Double: Dim d As Double
        'Define incomplete gamma integral parameters.
        a = delta / lambda: b = lambda: c = CVariable: d = delta / lambda
        'Calculate value of variable alpha.
        alpha = -c / b
        'Calculate upper incomplete gamma function lower bound.
        xm = IIf(m = 0, a, a * Exp(-lambda * m))
        'Calculate upper incomplete gamma function lower bound.
        xn = IIf(n = 0, 0, a * Exp(-lambda * n))
        'Calculate return variable value.
        mValue = Exp(d) * a ^ (c / b) * b ^ -1
        mValue = mValue * (UpperIncompleteGamma(alpha, xn) - UpperIncompleteGamma(alpha, xm))
    Else
        'Declare calculation variables.
        Dim x As Double: Dim lower As Double: Dim upper As Double
        'Define calculation variable values.
        x = omega_lt - kappa
        lower = IIf(m = 0, 1, Exp(x * m))
        upper = IIf(n = 0, 0, Exp(x * n))
        mValue = x ^ -1 * (upper - lower)
    End If
    'Return value to caller.
    EXPINT = mValue
End Function

'***************************
```

```
'Toolbox
'****************************

'Name: CTODCalc
'Purpose: Convert continuous-time rate to a discrete-time rate.
Public Function CTODCalc(continous_time_rate As Double) As Double
    CTODCalc = Exp(continous_time_rate) - 1
End Function

'Name: DTOCCalc
'Purpose: Convert discrete-time rate to a continuous-time rate.
Public Function DTOCCalc(discrete_time_rate As Double) As Double
    DTOCCalc = LN(1 + discrete_time_rate)
End Function

'Name: GammaDIST
'Purpose: Returns excel GammaDist function value
Public Function GammaDIST(x As Double, alpha As Double) As Double
    GammaDIST = Application.WorksheetFunction.GammaDIST(x, alpha, 1, True)
End Function

'Name: GammaLN
'Purpose: Returns excel GammaLn function value
Public Function GammaLN(value As Double) As Double
    GammaLN = Application.WorksheetFunction.GammaLN(value)
End Function

'Name: LambdaCalc
'Purpose: Return estimate of model parameter lambda.
Public Function LambdaCalc(half_life As Double) As Double
    LambdaCalc = -LN(0.5) / half_life
End Function

'Name: LowerIncompleteGamma
'Purpose: Returns lower incomplete gamma value.
Public Function LowerIncompleteGamma(alpha As Double, upper_bound As Double)
    'Declare calculation variables.
    Dim mValue As Double
    Dim mValue01 As Double
    Dim mValue02 As Double
    'Define calculation variable values.
    mValue01 = GammaDIST(upper_bound, alpha)
    mValue02 = GammaLN(alpha)
    mValue = mValue01 * Exp(mValue02)
    'Return value to caller.
    LowerIncompleteGamma = mValue
End Function

'Name: LN
'Purpose: Returns natural log of [value].
Public Function LN(value) As Double
    LN = Application.WorksheetFunction.LN(value)
End Function

'Name: MRRate
'Purpose: Return value of mean-reverting rate at time t.
```

```
Public Function MRRate(lambda As Double, rate_st As Double, rate_lt As Double, t As Double)
    'Declare calculation variables.
    Dim mValue As Double
    Dim delta As Double
    'Define calculation variable values.
    delta = rate_st - rate_lt
    mValue = rate_lt + delta * Exp(-lambda * t)
    'Return value to caller.
    MRRate = mValue
End Function

'Name: UpperIncompleteGamma
'Purpose: Returns upper incomplete gamma value.
Public Function UpperIncompleteGamma(alpha As Double, lower_bound As Double)
    'Declare calculation variables.
    Dim mValue As Double
    Dim mValue01 As Double
    Dim mValue02 As Double
    'Define calculation variable values.
    mValue01 = LowerIncompleteGamma(alpha, lower_bound)
    mValue02 = GammaLN(alpha)
    mValue = Exp(mValue02) - mValue01
    'Return value to caller.
    UpperIncompleteGamma = mValue
End Function
```